Приложение
Листинг файла script.kix
; Установление соединения с Active Directory
$rootDSE_ = GetObject("LDAP://RootDSE")

$domain_ = "LDAP://" + $rootDSE_.Get("defaultNamingContext")

$strADSQuery = "SELECT keywords, name, description, cn, uncname FROM '" +$domain_+"'  WHERE objectClass='volume'"

$objConnection = CreateObject("ADODB.Connection")

$objCommand = CreateObject("ADODB.Command")

$objConnection.CommandTimeout = 120

$objConnection.Provider = "ADsDSOObject"

$objConnection.Open ("Active Directory Provider")

$objCommand.ActiveConnection = $objConnection

$objCommand.CommandText = $strADSQuery

$st = $objCommand.Execute

$st.Movefirst

; Удаление подключенных сетевых дисков, которые обрабатываются скриптом
$i=0

$temp=""

Do 

$ds_s=""


$dss=$St.Fields("description").Value



for each $ds in $dss



$ds_s=$ds_s + $ds



Next

if len($ds_s)=1

use $ds_s+":" /delete /persistent

endif

$st.MoveNext

Until $st.EOF
; Определение членств пользователя в группах безопасности

dim $usergroup_name[]

$k=0

$i=0

DO


$Group = ENUMGROUP($i)




$ug1_len=len($group)-instrrev($group,"\")



redim preserve $usergroup_name[$k]



$usergroup_name[$k]=right($group,"$ug1_len")



$k=$k+1



$i=$i+1

UNTIL Len($Group) = 0


$st.Movefirst

; Подключение сетевых дисков
Do 

$cn=$St.Fields("cn").Value

$uncname=$St.Fields("uncname").Value

$ds_s=""


$dss=$St.Fields("description").Value



for each $ds in $dss



$ds_s=$ds_s + $ds



Next

$temp = $temp+$ds_s+"  "+$uncname+chr(13)

$group_b=right($cn,len($cn)-instrrev($cn,"."))

for $t=0 to ubound($usergroup_name)

if instr($usergroup_name[$t],$group_b)<>0 

if instr(ucase($group_b),ucase("PersonalGroup1"))<>0  or instr(ucase($group_b),ucase("PersonalGroup2"))<>0  or instr(ucase($group_b),ucase("PersonalGroup3"))<>0 


if instr($cn,@userid)<>0

use $ds_s+":" $uncname

? " !!!!" + $ds_s+":" $uncname



if @error=5 ; ошибка доступа к ресурсу



$error_code="1"






$error_message=$error_message+"Не удалось подключить диск "+$ds_s+" к ресурсу "+ $cn+chr(13)







EndIf


endif

else

use $ds_s+":" $uncname

? " !!!!" + $ds_s+":" $uncname



if @error=5 ; ошибка доступа к ресурсу



$error_code="1"






$error_message=$error_message+"Не удалось подключить диск "+$ds_s+" к ресурсу "+ $cn+chr(13)







EndIf

endif

endif

next


$st.MoveNext

Until $st.EOF

; Пауза
$delay_size=3

$Q = @Ticks + $delay_size*1000 ; x – количество секунд задержки

Do

Until @TICKS >= $Q

; Смена описания сетевых дисков в папке «Мой Компьютер»
$st.Movefirst

Do 

$uncname=$St.Fields("uncname").Value

$ds_s=""


$dss=$St.Fields("description").Value



for each $ds in $dss



$ds_s=$ds_s + $ds



Next

$keyw_s=""


$kss=$St.Fields("keywords").Value



for each $ks in $kss



$keyw_s=$keyw_s+cstr($ks)



Next

if len($keyw_s)>0

If  instr(ucase(@producttype), ucase("xp"))<>0 

$keyw_path="HKEY_CURRENT_USER\Software\Microsoft\Windows\CurrentVersion\Explorer\MountPoints2"

$uncname=Right($uncname,Len($uncname)-2)

$keyw_array=split($uncname,"\")

$temp1="#"

For Each $ugu in $keyw_array

$temp1=$temp1+"#"+$ugu

Next 

WriteValue(cstr($keyw_path+"\"+$temp1), "_LabelFromReg",cstr($keyw_s),"REG_SZ")

EndIf

if instr(ucase(@producttype), ucase("2000"))<>0 

$keyw_path="HKEY_CURRENT_USER\Software\Microsoft\Windows\CurrentVersion\Explorer\MountPoints"

$t=""

For $c=1 to Len($keyw_s)

$string=ucase(left(cstr(dectohex(Asc(Right(Left($keyw_s,$c),1)))),1))

$string2= right(cstr(dectohex(Asc(Right(Left($keyw_s,$c),1)))),1)


$flag_S=0


Select 

Case instr(ucase($string),UCase("c"))<>0

$t=$t+cstr("1")+$string2


$flag_S=1

Case instr(ucase($string),UCase("d"))<>0

$t=$t+cstr("2")+$string2
$flag_S=1

Case instr(ucase($string),UCase("e"))<>0

$t=$t+cstr("3")+$string2
$flag_S=1

Case instr(ucase($string),UCase("f"))<>0

$t=$t+cstr("4")+$string2
$flag_S=1

Case instr(ucase($string),UCase("g"))<>0

$t=$t+cstr("5")+$string2
$flag_S=1


EndSelect

 
If $flag_S=0 


$t=$t+cstr(dectohex(Asc(Right(Left($keyw_s,$c),1))))


endif

If Asc(Right(Left($keyw_s,$c),1))>128

$t=$t+"04"

Else

$t=$t+"00"

endif

Next

for $r=1 to 32-len($keyw_s)

$t=$t+cstr("00")

Next

WriteValue($keyw_path+"\"+$ds_s+"\_LabelFromReg", "Cache", $t, "REG_BINARY")

$r_w1=ReadValue($keyw_path+"\"+$ds_s+"\_LabelFromReg","version")

WriteValue($keyw_path+"\"+$ds_s+"\_LabelFromReg", "Version", $r_w1+1, "REG_DWORD") 

$r_w2=ReadValue($keyw_path+"\"+$ds_s+"\_GVI","version")

WriteValue($keyw_path+"\"+$ds_s+"\_GVI", "Version", $r_w2+1, "REG_DWORD") 

EndIf 

endif


$st.MoveNext

Until $st.EOF
